

Simo K. Kivelä, 28.10.2004

Kexleruksen viiniongelmia

Simon Kexlerus oli Suomen ensimmäinen matematiikan professori, joka työskenteli Turun akatemiassa sen perustamisesta lähtien vuosina 1640 – 1669. Seuraava probleema on häneltä peräisin:

Sinulla on viinejä, jotka maksavat 3, 5, 8 ja 10 markkaa pullolta. Ota yhteensä kymmenen täyttä pulloa ja tee niistä sekoitus, joka maksaa 6 markkaa pullolta. Montako pulloa kutakin viinilajia on otettava?

Ratkaisu

Pullojen määrät olkoot a , b , c ja d . Tällöin tulee seuraavien ehtojen olla voimassa:

```
In[1]:= pullot = a + b + c + d == 10
```

```
Out[1]= a + b + c + d == 10
```

```
In[2]:= rahat = 3 a + 5 b + 8 c + 10 d == 60
```

```
Out[2]= 3 a + 5 b + 8 c + 10 d == 60
```

On etsittävä ne ei-negatiiviset kokonaisluvut, jotka toteuttavat yhtälöt. Selvää on, että mahdolliset luvut ovat välillä $[0, 10]$.

Ehto funktiomuotoisena:

```
In[3]:= pullotJaRahat[{a_, b_, c_, d_}] = pullot && rahat
```

```
Out[3]= a + b + c + d == 10 && 3 a + 5 b + 8 c + 10 d == 60
```

Yksinkertaisin ohjelmointiteknikalla toteutettava ratkaisu on tarkistaa yhtälöiden toteutuminen kaikkien vaihtoehtojen kohdalla. Vaihtoehtoja on

```
In[4]:= 11 ^ 4
```

```
Out[4]= 14 641
```

kappaletta.

Siis:

```
In[5]:= ratkaisu = Select[Flatten[Table[{a, b, c, d}, {a, 0, 10},  
    {b, 0, 10}, {c, 0, 10}, {d, 0, 10}], 3], pullotJaRahat]
```

```
Out[5]= {{0, 8, 0, 2}, {1, 5, 4, 0}, {2, 4, 3, 1},  
    {3, 3, 2, 2}, {4, 0, 6, 0}, {4, 2, 1, 3}, {5, 1, 0, 4}}
```

Mahdollisia ratkaisuja on siis 7 kappaletta.

Nykyisillä ohjelmistoilla tällainen ratkaiseminen ei kestä kauan. Kexleruksen aikana tilanne oli toinen ja tehtävälle on toki löydettävissä varsin yksinkertainen ratkaisu päästelemällä. Tämänkin voi toteuttaa Mathematicalla:

Perinteisempi ratkaisu

Ratkaistaan pulloja ja rahoja koskevista ehdoista esimerkiksi a ja b :

```
In[6]:= rtk = Solve[{pullot, rahat}, {a, b}]
```

```
Out[6]= {{a -> 1/2 (-10 + 3 c + 5 d), b -> 15 - 5 c/2 - 7 d/2}}
```


Koska kaikkien tuntemattomien tulee olla ei-negatiivisia, saadaan seuraavat ehdot:

```
In[7]:= ehdot = a >= 0 && b >= 0 && c >= 0 && d >= 0 /. rtk[[1]]
```

```
Out[7]= 1/2 (-10 + 3 c + 5 d) >= 0 && 15 - 5 c/2 - 7 d/2 >= 0 && c >= 0 && d >= 0
```

Piirretään alue, jossa ehdot toteutuvat:

```
In[8]:= alue = RegionPlot[ehdot, {c, 0, 10},  
  {d, 0, 10}, GridLines -> {Range[0, 10], Range[0, 10]}]
```



Ne alueen sisään osuvat cd -tason kokonaislukupisteet, joita vastaavat a ja b myös ovat kokonaislukuja, antavat tehtävän ratkaisut. Kokeiltavia vaihtoehtoja on siis varsin rajallinen määrä.

Tarkistuksena edellisessä ratkaisussa saadut pisteet lisättyinä kuvioon:

```
In[9]:= Show[alue, Graphics[Map[Point[Drop[#, 2]] &, ratkaisu]]]
```

